Algorytm adaptacyjnej filtracji LMS

Filtracja adaptacyjna polega na nieustannym dostrajaniu się układu filtrującego do zmiennych warunków pracy, to jest zmiennego w czasie charakteru zakłócenia, które dany układ ma za zadanie eliminować.

Najpopularniejszym algorytmem wykorzystywanym w filtracji adaptacyjnej jest algorytm LMS (ang. Least Mean Square).

Rozważmy przykład zastosowania metody filtracji adaptacyjnej, który pomoże dokładnie zobrazować użyteczność tej techniki w praktycznych urządzeniach elektronicznych powszechnego użytku.

Rysunek 1 przedstawia schemat układu filtracji adaptacyjnej LMS sygnału mowy w obecności sygnału zakłócającego. Może to być przykład kierowcy rozmawiający podczas jazdy przez telefon komórkowy korzystając z zestawu głośnomówiącego. W takim przypadku mikrofon rejestruje jego mowę oraz zmieszany z nią sygnał zakłócenia (d(n)). Sygnał zakłócenia (x(n)) stanowić może na przykład szumu otoczenia lub warkot silnika. Jeżeli teraz w samochodzie znajdzie się dodatkowy mikrofon odniesienia rejestrujący tylko hałas zakłócający – to zastosowanie filtracji adaptacyjnej pozwoli usunąć to zakłócenie z sygnału mowy.

Oczywiście zakłócenie rejestrowane przez główny mikrofon i obecne w sygnale mowy oraz zakłócenie rejestrowane przez mikrofon odniesienia różnią się od siebie (jeżeli chodzi o amplitudę i fazę), ponieważ oba mikrofony znajdują się w innych miejscach. Oba sygnały są jednak w pewien sposób ze sobą skorelowane i ta korelacja właśnie jest wykrywana przez algorytm LMS, a następnie wykorzystywana do usunięcia składowej zakłócającej z sygnału mowy kierowcy.

Zakłócenie generowane przez silnik może mieć zmienny charakter, który jakkolwiek nie ma znaczenia w przypadku poprawnego działania układu filtra adaptacyjnego, ponieważ filtr nieustannie dostosowuję się do nowych warunków pracy.

Rys. 1. Schemat blokowy układu filtracji adaptacyjnej LMS.

Rys. 2. Filtr FIR z przestrajanymi współczynnikami „h” zastosowany w algorytmie filtracji adaptacyjnej LMS.

W algorytmie LMS strukturą dokonującą zasadniczej filtracji jest filtr o skończonej odpowiedzi impulsowej FIR (ang. Finite Impulse Response) (rysunek 2).Zasada jego działania polega na przechowywaniu w rejestrze przesuwnym odpowiedniego wektora zarejestrowanych w danym odcinku czasu próbek, które następnie wymnażane są przez stosowne współczynniki „h” i sumowane. 

Współczynniki filtru FIR nie są stałe, są one zmienne w czasie i dostrajają się tak, aby sygnał y(n) dążył do osiągnięcia stanu, w którym byłby on odpowiednikiem zakłócenia znajdującego się w sygnale d(n). Po operacji odjęcia go od tego sygnału w sygnale błędu e(n) powinna znaleźć się w miarę „czysta” mowa kierowcy.

Współczynniki filtra „h” w zastosowanym algorytmie filtracji adaptacyjnej LMS zmieniają się z każdą mowo zarejestrowaną próbką. Do aktualnych wartości wektora „h” dodawana jest pewna korekta wyliczona przez algorytm LMS i tak zmodyfikowane wartości stają się w następnym cyklu próbkowania aktualnymi współczynnikami biorącymi udział we właściwej filtracji.

Wektor macierzowy H(n) zawiera zbiór kolejnych wartości współczynników „h” w chwili próbkowania n. Wartość macierzy H(n+1) w następnej chwili próbkowania wyznaczana jest na podstawie aktualnej macierzy H(n) zmodyfikowanej korektą wyliczoną przez adaptacyjny algorytm LMS.

Omówienie praktycznej implementacji dla ADSP21161 w trybie symulatora

Poniżej znajduje się opis przykładowego kodu niskiego poziomu napisany na procesor sygnałowy SHARC 21161 firmy Analog Devices. Przykład ten zawiera wszystkie kluczowe struktury inicjalizujące zarówno segment danych jak i segment programu odpowiednimi danymi, a także instrukcje inicjalizujące generatory adresów DAG ( także bufory cyrkularne zawarte w  strukturze procesora) które to znajdują się w pliku o nazwie fir_blk_test.asm. Sama procedura filtracji oraz obliczeń i korekcji nowych wartości współczynników „h” dokonywana przez algorytm filtracji adaptacyjnej LMS (Least Mean Square, tj. minimalizacji błędu średniokwadratowego) została zawarta w pliku fir_blk.asm. 

Program realizowany jest w architekturze nieskończonej pętli, w której wszystkie sygnały wejściowe pochodzą z plików i są regularnie wczytywane w sposób właściwy adresowaniu buforów kołowych, tzn. po przeniesieniu wskaźnika adresu w pliku spod którego czytana jest dana następuje ustawienie wskaźnika adresu w pliku na początek pliku. W celu zaimplementowania tego programu w sprzęcie (EZ Kit Lite ADSP21161) należy dodać do programu kody inicjalizujące i konfigurujące przetworniki sygnału, interfejsy wymiany danych, dostęp do pamięci itp.

Z kodu programu zawartego w pierwszym pliku dokonywany jest skok bezwarunkowy do kodu zawartego w drugim pliku zawierającego właściwy algorytm adaptacyjny typu LMS. Dzieje się tak ze względu na to że w projekcie nie ma syustemu zarządzającego pobieranie próbki z przetworników i wykonywaniem odpowiednich czynności obsługujących sprzęt i interfejsy służące do wymiany danych.

 Algorytm programu zawartego w pliku jest bardzo prosty i składa się z dwóch pętli, z których pierwsza odpowiada za filtrację cyfrową, a druga aktualizuje współczynniki filtru cyfrowego na podstawie sygnału błędu.

Kolejność użycia pętli jest bardzo istotna ponieważ pętla druga korzysta z danych będących wynikiem działania pętli pierwszej. Do danych tych należy próbka sygnału wyjściowego – odpowiedź filtru cyfrowego i obliczony na jej podstawie sygnał błędu (będący różnicą wartości próbki sygnału referencyjnego i próbki będącej wynikiem filtracji próbki sygnału zakłócającego).

Pętla druga korzysta z tych danych w celu modyfikacji współczynników filtru adaptacyjnego. W projekcie maksymalnie wykorzystano możliwości generatorów adresu tak aby nie trzeba było zwiększać kodu o instrukcje potrzebne do modyfikacji poszczególnych rejestrów adresowych.
Instrukcja do ćwiczenia

1. Uruchomić VisualDSP++ i okno sesji w którym wybierz procesor ADSP-21161 i tryb symulacji.

2. Otwórz projekt 21161_adapt i prześledź zawartość okna oraz zapoznaj się z plikami. Czy rozumiesz co oznaczają poszczególne dyrektywy i instrukcje? Które instrukcje odpowiadają za „wczytanie” próbek sygnałów do rejestrów roboczych? (bufor „dline” – linia opóźniająca ze wskaźnikami i rejestrami w DAC). Które instrukcje dokonują filtracji?  Które obliczają sygnał błędu i modyfikują współczynniki filtru? Która pętla odczytuje próbki (od aktualnej do N-1) i kolejne współczynniki? Które instrukcje ustawiają adres bazowy wskaźnika adresu na odpowiednią komórkę w pamięci? Gdzie następuje zapis wyników filtracji do bufora wyjściowego (output)? Szczególną uwagę zwróć na pliki dołączone w formacie dat, zawierać one będą współczynniki filtru FIR i próbki sygnału referencyjnego – odniesienia, a także sygnału zakłócającego, który zostanie poddany filtracji adaptacyjnej.

3. W celu zmiany plików dat na inne dokonaj następujących kroków:

· zaznacz istniejące pliki w formacie dat i usuń je z bieżącego projektu

· otwórz wybrane przez Ciebie pliki w formacie dat i dodaj je do projektu

· uaktualnij cały projekt wybierając z menu opcje: Project->Update Dependencies

· następnie dokonaj odpowiednich zmian w pliku fir_blk-test.asm zawierającym kod inicjalizujący wpisując w odpowiednie pola nazwy plików w formacie dat zawierające próbki sygnałów
· wybierz w tym samym pliku odpowiednią wartość liczbową dla współczynnika mju odpowiadającego za szybkość zbieżnośći procesu adaptacyjnego

· zapisz wszystkie dokonane zmiany oraz ponownie wybierz opcję Project->Update Dependencies

· dokonaj kompilacji całego projektu wybierając opcję Project->Build Project lub użyj odpowiedniej ikony

4. 
W celu obserwacji rezultatów działania programu wybierz z menu View->Debug Windows->Plot->New umieszczając na rysunkuzawartości odpowiednich rejestrów kołowych zawierających: próbki sygnału referencyjnego, zakłóconego, współczynniki, próbki sygnału wyjściowego i próbki sygnału błędu, wpisując w odpowiednich polach właściwe wartości:

Title: wpisz tytuł generowanego wykresu

Data Setting: 

Pole: Name – wpisz etykietę danych pod jaką b ędą widoczne na legendzie wykresu

Pole: Memory – wybierz segment w którym umiejscowione są dane (przyciśnij Browse aby wybrać jednostkę zmiennych w której zawarte są właściwe dane, to spowoduje że w okienku ADDRESS pojawi się nazwa odpowiedniej zmiennej)

Pole Count – wybierz długość bufora danych, powinna się ona zgadzać z wartością zadeklarowaną wcześniej w pliku fir_blk_test.asm

Pole Data: Pamiętaj o wyborze odpowiedniego formatu danych

Na zakończenie przyciśnij przycisk ADD tak aby umiejscowić dane na wykresie.

Podobnie postępuj z innymi danymi zawierającymi wartości interesujących Ciebie sygnałów.

Wynikiem tego działania powinno być okno graficzne zawierające konfigurowany przez Ciebie wykres. W celu poprawy czytelności dodaj do wykresu legendę.

5. Zacznij wykonywać program w trybie krokowym śledząc wyniki na wykresie. Możesz umieścić w odpowiednich miejscach programu tzw. pułapki tj. Breakpoint’y i wykonywać program w trybie Run – Halt.

6. Po wykonaniu powyższych kroków powróć do pliku fir_blk_test.asm i zmień wartość współczynnikia mju i zaobserwuj efekty.

/**************************************************************

File Name:
fir_blk_test.asm

Purpose:
Program to call fir_blk.asm

**************************************************************/

#include    "def21161.h"

//#define     TAPS    64

/* length of filter */

//#define     N       128

/* number of samples */

#define     TAPS    256

/* length of filter */

#define     N       256

.extern
fir;

/* DM data */

.section/dm seg_dmda;

.ALIGN 2;

.var    dline[TAPS];


/* delay line compensate for circ buffer, see comments in block_fir.asm */

.ALIGN 2;

.var    input[N] = "y1.dat";
/* array of samples */

.ALIGN 2;

.var input2[N] = "y.dat";

/* PM data */

.section/dm seg_pmda;

.ALIGN 2;

.var    coeffs[TAPS] = "wsp256.dat";

.ALIGN 2;

.var    output[N];/* Output array.  The first and last entries are dummy */



    /* entries to allow an optimization in the fir.*/

.ALIGN 2;

.var e[N];

.Align 2;

.var mju=0.001;

/* PM interrupt vector code */

.section/pm   seg_rth;

Reserved_1:     rti; nop; nop; nop;

Chip_Reset:     idle; jump start; nop; nop;

/* program memory code */

.section/pm seg_pmco;

/* Initialization code */

start:

      b0 = dline;

      l0 = @dline;


  b1 = input;


  l1 = @input;


  b2 = dline;


  l2 = @dline;


  b3 = input2;


  l3 = @input2;


  b4 = dline;


  l4 = @dline;


  b7 = dline;


  l7 = @dline;


  b8 = coeffs;


  l8 = @coeffs;


  b9 = output;


  l9 = @output;


  b10 = coeffs;


  l10 = @coeffs;


  b11 = coeffs;


  l11 = @coeffs;


  b14 = e;


  l14 = @e;


  b15 = mju;


   bit set MODE1 CBUFEN;


  //m0=0;


 // m1=1;


 // m2=-1;


 // m3=2;


  m0=0;


  m1=1;


  m2=-1;


  m3=2;


  m4=-2;


  m8=-1;


  m9=2;


  m10=1;


  m11=-2;


  m12=0;


  r1=N/2;


  f6=pm(i15,m12);


  r2=(N-2)/2;


  wait1: 
  


  call fir (db);


  f9 = 0;


  s6=f6;

/* Terminate and wait */

  jump wait1;

/**************************************************************

File Name:
fir_blk.asm

Purpose:
Subroutine that implements a Block FIR Filter given




coefficients and samples.

Equation:
y(n) = Summation from k=0 to M of h(k)*x(n-k)

Calling Parameters:




b0,i0 = address of delay line buffer




l0 = length of delay line buffer




b1,i1 = address of input samples buffer




b8,i8 = address of coefficients buffer




l8 = length of coefficients buffer




b9,i9 = address of output buffer




r1 = number of taps in the filter divided by 2




r2 = number of samples




r3 = (number of taps in filter - 6) / 2




s0,m0,l1,l9 = 0




m1,m10 = 1




m2=-1




m3,m9 = 2

Assumptions:
All arrays must start on even address boundaries.




All arrays must have an even number length (zero pad if necessary)

Return Values:
i9=OUTPUT

Registers Affected:
f0,s0,f4,s4,f8,s8,f12,s12,i0,i8,i9

Cycle Count:
9 + taps/2 + samples(6 + taps/2) + 9 cache misses

Number of PM Locations:




21 instruction words




2 * Number of taps locations for coefficients




Number of samples + 1 locations for the output buffer

Number of DM Locations:




Number of taps locations for the delay line buffer




Number of samples locations for the input buffer

Circular buffer notes: 

Because SIMD or Long word access transfer two 32-bit words, programs must be 

careful when using these accesses in circular buffering.  It is important that 

SIMD or Long word accesses do not cross a circular buffer boundary.  If a SIMD 

mode access occurs using a circular buffer index register that points to the 

last location in the circular buffer (end of buffer), the resulting access 

transfers the last location in the circular buffer and the 

first location outside the buffer (end of buffer + 1).  

**************************************************************/

#include    "def21160.h"

.global fir;

/* program memory code */

.section/pm seg_pmco;

fir:

    bit set MODE1 CBUFEN | PEYEN;

    nop;

    f9 = dm(i3,m3);

    f9 <-> s9;

    f5 = f5-f5,f0 = dm(i1,m3);






    f10 = f9;

    s9 <-> f10;

    f0 <-> s0;

    dm(i0,m3) = f0;

    f11 = f11-f11,f0 = dm(i2,m4),f4 = pm(i8,m9);

 lcntr = r2, do main_fir until lce; 

     f5 = f0*f4,f0 = dm(i2,m4),f4 = pm(i8,m9);

main_fir:    f11= f11+f5;

    f5 = f0*f4;

    f11 = f11+f5; 

    f8 = s11;

    f11 = f11+f8;

    f8  = f11;

    s11 = f8;

    f9  = f9-f11;

    pm(i14,m10)=f9; 

    f9=f9*f6;

lcntr = r1, do adapt until lce;   

    f12 = pm(i10,m9),f3 = dm(i4,m4); 

    f3 = f3*f9;

    f12 = f12+f3;

adapt:    pm(i11,m9) = f12;

        modify(i4,m3);

        modify(i2,m3);

bit clr MODE1 PEYEN;



nop;



pm(i9,m9)=f11;



rts;

fir.end:

